GBT Memo No. 44

DISTORTIONS OF THE ANTENNA PATTERN DUE TO SURFACE-PANEL IMPERFECTIONS

FRED SCHWAB

March 18, 1990

GBT Memoranda No. 28 and No. 35 address the question of the least number of precision molds that might be required for fabrication of panels for the primary surface of the Green Bank Telescope. The previous memos describe the patterns of systematic surface errors that would arise if panels were molded to match slightly wrong portions of the design paraboloid; however, these memos do not investigate the electromagnetic consequences. Here I take a look at the latter concern and, in particular, describe the distortions of the antenna pattern, as calculated via one-dimensional simulations.

I have written a Fortran program to calculate the radiation power pattern of a uniformly-illuminated one-dimensional aperture with varying wavefront phase distortions across the aperture (see Appendix). These phase distortions follow the pattern of systematic surface irregularities corresponding to the 13-mold and 20-mold examples of Memo 35, along a line across the middle of the dish—from a point near the vertex (of the parent paraboloid) to the far edge of the reflector. The assumed panel geometry is illustrated in Figure 1, and the surface error along the line chosen for analysis is shown in Figure 2.

The surface errors $\Delta\delta$ shown in Figure 2 are measured in directions normal to the design paraboloid. At the vertex of the paraboloid, the wavefront phase error (the sum of the phase-path errors from celestial radio source to main reflector and from reflector to focus) would be equal to $4\pi\Delta\delta/\lambda$ radians, while at the far edge of the reflector, at $r = 104$ m, the phase error would be smaller by a factor $\approx 3/4$. My program ignores this difference; it samples the surface errors at 2048 points (about fifty points per panel) and takes the phase distortion $\Delta\varphi$ to be the larger number, $4\pi\Delta\delta/\lambda$, all across the aperture.

The power pattern is calculated by computing the squared modulus of the discrete Fourier transform of the sequence of 2048 unit exponentials $e^{i\Delta\varphi}$ (extra zeros are appended to this sequence, in order to get several data points across each sidelobe, so that none of the peaks are underestimated—and for a more pleasing display). The surface efficiency η is computed by dividing the power in the main lobe of the beam by the power that would be present in the absence of any phase distortions. In its printed output, the program compares this calculated surface efficiency with the efficiency that would be predicted by Ruze' formula, $e^{-(4\pi\sigma/\lambda)^2}$.[1]

The severest distortion of the antenna pattern would occur at the uppermost observing frequency. Plots of the beam at an observing frequency of 115 GHz are shown in Figure 3 (the near-in sidelobes) and Figure 4 (the far-sidelobe pattern). Both sides of the beam pattern are shown, since the beam inherits some asymmetry

[1] This is done mainly as a cross-check of the calculations. In all cases there has been good agreement.

from the asymmetry of the aperture phase distortion. The abscissae θ of these plots are labeled in units of λ/D (essentially, beamwidths). The most prominent sidelobes occur when θ is equal to an integer multiple of the number of molds. (Along with each of the plots of the antenna pattern is shown the envelope of $\mathrm{sinc}^2\,\theta$, the beam pattern that corresponds to a uniformly-illuminated aperture with no phase errors.)

Figures 5 and 6, again calculated for an observing frequency of 115 GHz, show the effect on the antenna pattern of adding to the systematic surface errors an additional 75 μm r.m.s. component of spatially uncorrelated random error. The innermost sidelobes due to the systematic errors are still quite evident, but the far-sidelobes due to these errors are much less prominent. The antenna pattern corresponding to the random component alone is shown in Figure 7.

Figures 8 and 9 show the antenna patterns (taking into account only the systematic surface errors) for the 13- and 20-mold cases at a factor-of-ten lower observing frequency—11.5 GHz. Figure 10 shows the effect of again adding in a 75 μm r.m.s. component of random error (for the 20-mold case only); the innermost few sidelobes due to the systematic errors are still evident, but they are down more than 40 dB below the main lobe.

Finally, Figures 11 and 12 show the effect, at 115 GHz, of random panel-setting errors. Each panel offset is in error by some amount in the range ± 150 μm; the panel-mold errors correspond to the 20-mold case. Some of the near sidelobes exceed 1%, the sidelobes due to the imperfect molds are very prominent, and, in particular, those near $\theta = \pm 20$ have been enhanced. A more realistic simulation of setting errors would include panel tilt and spatial correlation of the errors.

Discussion

The approach I have taken could be used for further studies of the electromagnetic effects of panel imperfections, panel setting errors, etc. Additional one-dimensional simulations might be useful, and full-scale two-dimensional simulation and more realistic modeling of these effects might make for an interesting summer-student project.

It would be straightforward to do the two-dimensional simulations, and we have computer resources adequate for the task (capabilities for large, 2-D FFTs, and good display software, within AIPS)—but it would require considerable effort: probably a few weeks to simulate and analyze actual panel layouts; or less time to treat a simple rotational geometry (the case of a symmetric antenna aperture could done via a few numerically-computed Fourier–Bessel transforms).

However, I believe that one-dimensional simulations probably can provide enough insight into the panel-mold question. After a discussion with Rick Fisher, I believe that the effect of the actual panel geometry will be to diminish the spurious sidelobe levels below the levels indicated by the one-dimensional simulations.

Conclusion

The electromagnetic consequences of reducing the number of surface-panel molds—perhaps even drastically, to as few as thirteen—do not appear to me to be too severe. We should solicit the opinions of potential users of the GBT to see whether they concur.



Figure 1. The panel geometry used for one-dimensional simulations of the effect of systematic errors in the primary-surface panels of the GBT. The contour plots, reproduced from Figure 1 of GBT Memo No. 35, represent the pattern of systematic errors over two of the surface panels, in accord with the 13-mold calculation of the earlier memo. The patterns of error over the other surface panels along this row would be qualitatively similar to the two shown here, except that in the 13-mold scheme every third panel would be a precise match to the design paraboloid.

The one-dimensional aperture that was used for the beam-pattern calculations is the line along the lower boundary of this row of panels. The surface error along this line is shown in Figure 2.



Figure 2. The surface error, measured normal to the design paraboloid, along the line used for one-dimensional simulations. *(Top)* The error pattern corresponding to the 13-mold scheme (Table 2 of Memo 35); *(Bottom)* the error pattern corresponding to the 20-mold scheme (Table 3 of Memo 35).



Figure 3. Plots of the power radiation pattern of the 1-D aperture at 115 GHz for the 13- and 20-mold schemes. Here the near sidelobes are shown. The abscissae are labeled in units of λ/D, so that $\theta = 1$ corresponds to $\sim 5''\!.4$. The smooth curves represent the envelope $1/(\pi\theta)^2$ of $\mathrm{sinc}^2\,\theta$; i.e., the envelope of the beam pattern that would correspond to a uniformly-illuminated one-dimensional aperture D/λ wavelengths in length, without phase errors. The far-sidelobe behavior is shown in Figure 4.





Figure 4. Like Figure 3, except showing the far-sidelobe behavior for the 13- and 20-mold schemes, at 115 GHz. The beam pattern is asymmetric because the phase distortions across the aperture that are caused by surface imperfections are not symmetric about the center of the aperture.



Figure 5. Like Figure 3, showing the near-sidelobe behavior at 115 GHz, except with 75 μm r.m.s. random error (spatially uncorrelated, zero mean, and normally distributed) added to the systematic surface errors. The far-sidelobe behavior is shown in Figure 6.



Figure 6. Like Figure 4, showing the far-sidelobe behavior at 115 GHz, except with 75 μm r.m.s. random error added to the systematic surface errors.



Figure 7. The near- and far-sidelobe behavior at 115 GHz for purely random (75 μm r.m.s.) surface errors. Compare with Figures 5 and 6.



Figure 8. The near-sidelobe patterns at 11.5 GHz, due to systematic panel errors, corresponding to the 13- and 20-mold schemes. $\theta = 1$ corresponds to an offset (λ/D) of $\sim 54''$. Compare with Figure 3.



Figure 9. The far-sidelobe patterns at 11.5 GHz, due to systematic panel errors, corresponding to the 13- and 20-mold schemes. Compare with Figure 4.



Figure 10. The sidelobe pattern at 11.5 GHz corresponding to the 20-mold scheme, with 75 μm r.m.s. random error added to the systematic surface errors.



Figure 11. Surface errors, for the 20-mold case, with (spatially uncorrelated) random panel-setting errors of ± 150 μm included (see text). The corresponding radiation power pattern of the 1-D aperture, at 115 GHz, is shown in Figure 12.





Figure 12. Beam pattern, at 115 GHz, for the 20-mold case with random panel-setting errors included (see Fig. 11).

-15-

Appendix

(1)

```
      program syserr
c  Program to do a one-dimensional simulation of the effect that
c  systematic errors in surface panels of the GBT would have on the
c  beam pattern.  The program generates errors along a line through
c  the center of the dish, in accord with the mold utilization schemes
c  of GBT Memo 35, Tables 2 and 3.  The radiation power pattern of
c  a uniformly-illuminated one-dimensional aperture is calculated,
c  with phase errors in the aperture corresponding to a given observing
c  frequency and the assumed surface irregularities.

c  The FFT routine and the Gaussian random noise generator are from
c  the IMSL library.  The plotting routines fgraph and fgraphm, which
c  use the Caltech graphics package, are from my own library.
      implicit real*8 (a-h,o-z)
      dimension rcents(40),rdesigns(40),offsets(40),off20(40),seterr(40)
      data offsets/19.7,11.3,0,-14.2,19.6,0,-22.2,27.0,0,-29.2,
     & 33.2,0,-35.0,38.1,0,-39.4,41.5,0,-42.4,43.7,0,-44.2,44.8,
     & 0,-45.0,45.0,0,-44.8,44.3,0,-44.0,43.1,0,-42.6,41.5,0,-40.8,
     & 39.5,0,-38.8/
      data off20/3.8,-4.6,6.7,-7.4,9.5,-10.2,12,-12.6,14.4,-14.9,
     & 16.4,-16.8,18.1,-18.5,19.5,-19.8,20.7,-20.9,21.5,-21.6,
     & 22,-22.1,22.4,-22.4,22.5,-22.5,22.4,-22.4,22.2,-22.1,
     & 21.8,-21.7,21.4,-21.2,20.8,-20.7,20.2,-20,19.5,-19.3/
      real*4 errn(2048),x(2048),xp(5000),yp(5000,2)
      real*8 rwksp(196658)
      complex*16 z(16384)
      common rcent,rdesign,c,offset,rpsi,reta
      common/worksp/rwksp
      real*4 rnnof
      character*80 label
      pi=4d0*atan(1d0)
      c=60d0
      rpsi=1.25d0
      reta=1d0

      print *,'Type number of molds (13 or 20)'
      read *,nmolds
      print *,'Type observing frequency (GHz)'
      read *,freq
      print *,'Choose one:'
      print *,'1 -- Just include panel mold systematic error'
      print *,'2 -- The above, plus 75 micron random error'
      print *,'        with no spatial correlation'
      print *,'3 -- Just 75 micron pure random error, with no syst.
      print *,'4 -- No surface error'
      print *,'5 -- Panel mold error, plus setting error '
      print *,'        (+/- 75 microns)'
      print *,'6 -- Just setting error (+/- 75 microns)'
      read *,iopt

      if (nmolds.eq.13) then
         xx=-10.25d0
         do i=1,40
            rcents(i)=5.25d0+(i-1)*2.5d0
            if (mod(i,3).eq.2.and.i.ne.2) xx=xx+7.5d0
            rdesigns(i)=xx
            offsets(i)=1d-6*offsets(i)
            if (iopt.eq.5.or.iopt.eq.6) seterr(i)=2.*(rnunf()-.5)*75.
         end do
      else
```

(2)

```
         xx=-6.5d0
         do i=1,40
            rcents(i)=5.25d0+(i-1)*2.5d0
            rdesigns(i)=xx
            if (mod(i,2).eq.0) xx=xx+5d0
            offsets(i)=1d-6*off20(i)
            if (iopt.eq.5.or.iopt.eq.6) seterr(i)=2.*(rnunf()-.5)*75.
         end do
      end if

      iseed=133457
      call rnset(iseed)
      n=2048
      rms=0d0
      do k=1,n
         r=(k-.5d0)*100d0/n+4d0
         i=(r-4d0)/2.5d0+1d0
         psi=r-rcents(i)
         eta=reta
         rcent=rcents(i)
         rdesign=rdesigns(i)
         offset=offsets(i)
         if (iopt.eq.1) errn(k)=f(psi,eta)*1d6
         if (iopt.eq.2) errn(k)=f(psi,eta)*1d6+75d0*rnnof()
         if (iopt.eq.3) errn(k)=75d0*rnnof()
         if (iopt.eq.4) errn(k)=0d0
         if (iopt.eq.5) errn(k)=f(psi,eta)*1d6+seterr(i)
         if (iopt.eq.6) errn(k)=seterr(i)
         x(k)=r
         rms=rms+errn(k)**2
      end do
      rms=sqrt(rms/n)
      print *,'R.m.s. surface error (microns)=',rms

      encode (80,1,label) nmolds,rms
1     format(i2,' molds   rms=',f6.2,' microns')
      if (iopt.ne.4)
     &call fgraph(x,errn,n,6.,6.,10,10,.false.,.false.,
     &   .true.,.false.,.true.,.false.,0.,0.,
     &   'Radius, r','Surface Error (microns)',label)

      wavelength=.2997925d0/freq
      call iwkin(196658)
      nfft=16384
      do i=1,nfft
         if (i.le.n) then
            phaserr=2d0*errn(i)*1d-6/wavelength
            theta=2d0*pi*phaserr
            z(i)=dcmplx(cos(theta),sin(theta))
         else
            z(i)=0d0
         end if
      end do
      call dfftcf(nfft,z,z)
      eta=(abs(z(1))/2048d0)**2*1d2
      print *,'Surface efficiency (percent)=',eta
      rmsr=wavelength*sqrt(-log(eta/1d2))/4d0/pi*1d6
      print *,'R.m.s. error from Ruze''s formula (microns)=',rmsr

c  Plot of the far sidelobes (positive theta, only):
```

(1)

```
      nw=4400
      do i=1,nw
         xp(i)=(i-1)/8d0
c           Could opt to normalize to unity at beam center,
c           or to normalize to power in main lobe that would
c           correspond to no surface errors; I've chosen the
c           latter:
c        y=(abs(z(i))/abs(z(1)))**2
         y=(abs(z(i))/2048)**2
         if (y.eq.0d0) y=1d-30
         y=log10(y)
         yp(i,1)=y
         if (i.gt.1) yp(i,2)=log10(1./(pi*xp(i))**2)
      end do
      yp(1,2)=yp(2,2)
      if (iopt.eq.4) nmolds=40
      encode (80,2,label) nmolds,freq,eta
2     format(i2,' molds   f=',f5.1,' GHz','   eta=',f7.3,'%')
      call fgraphm(xp,yp,5000,2,nw,6.,6.,10,10,.false.,.true.,
     1   .true.,.false.,.false.,.true.,-8.,0.,
     2   'Theta (units of reciprocal aperture width)',
     3   'Normalized Power',label)
c
c  Plot of the near-in sidelobes, along with the envelope of
c  sinc^2(theta) (positive theta, only):
      nw=639
      do i=2,nw
         yp(i,2)=log10(1./(pi*xp(i))**2)
      end do
      yp(1,2)=yp(2,2)
      ncurv=2
      if (iopt.eq.4) ncurv=1
      call fgraphm(xp,yp,5000,ncurv,nw,6.,6.,10,10,.false.,.true.,
     1   .true.,.false.,.false.,.true.,-8.,0.,
     2   'Theta (units of reciprocal aperture width)',
     3   'Normalized Power',label)

      if (iopt.eq.4) stop
c  Plot of the far sidelobes (for both positive & negative theta):
      nw=2399
      do i=-nw,nw
         xp(i+nw+1)=i/8d0
         if (i.ge.0) then
            y=(abs(z(i+1))/2048)**2
         else
            y=(abs(z(nfft+1+i))/2048)**2
         end if
         if (y.eq.0d0) y=1d-30
         y=log10(y)
         yp(i+nw+1,1)=y
         yp(i+nw+1,2)=log10(1./(pi*max(1,abs(i))/8d0)**2)
      end do
      nw=2*nw+1
      call fgraphm(xp,yp,5000,2,nw,6.,6.,10,10,.false.,.true.,
     1   .true.,.false.,.false.,.true.,-8.,0.,
     2   'Theta (units of reciprocal aperture width)',
     3   'Normalized Power',label)
c
c  Plot of the near-in sidelobes, along with the envelope of
c  sinc^2(theta), (for both positive & negative theta):
```

-16-

(4)

```
      nw=639
      do i=-nw,nw
         xp(i+nw+1)=i/8d0
         if (i.ge.0) then
            y=(abs(z(i+1))/2048)**2
         else
            y=(abs(z(nfft+1+i))/2048)**2
         end if
         if (y.eq.0d0) y=1d-30
         y=log10(y)
         yp(i+nw+1,1)=y
         yp(i+nw+1,2)=log10(1./(pi*max(1,abs(i))/8d0)**2)
      end do
      nw=2*nw+1
      call fgraphm(xp,yp,5000,2,nw,6.,6.,10,10,.false.,.true.,
     1   .true.,.false.,.false.,.true.,-8.,0.,
     2   'Theta (units of reciprocal aperture width)',
     3   'Normalized Power',label)


      stop
      end

      double precision function f(psi,eta)
c  Tangent planes are located at two distinct radii, rcent and rdesign,
c  from the vertex of the paraboloid.  For points on the paraboloid at
c  the same tangent plane coordinates above the two planes, this function
c  subroutine calculates the difference in heights (possibly adding an
c  offset, if the variable 'offset' is nonzero).
c  The psi-axis runs in the direction of positive r at the tangent point,
c  and the eta-axis in the direction of positive azimuth, phi.
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,c,offset,rpsi,reta
      f=t(rcent,c,psi,eta)-t(rdesign,c,psi,eta)-offset
      return

      entry fsq
c  fsq is the square of f.
      fsq=(t(rcent,c,psi,eta)-t(rdesign,c,psi,eta)-offset)**2
      end

      double precision function t(r,c,psi,eta)
c  Given tangent plane coordinates (psi,eta) of a point on the
c  paraboloid, this function subroutine calculates the height of
c  that point above the tangent plane.  The point of tangency,
c  (psi,eta)=(0,0), is centered at a distance r=sqrt(x**2+y**2)
c  away from the vertex.  The height above the tangent plane is
c  measured along the direction of the surface normal at psi=eta=0.
      implicit real*8 (a-h,o-z)
      a=sqrt(r**2+4d0*c**2)
      t = (-a*sqrt(4d0*a*c*psi*r+(4d0*c**2-a**2)*eta**2+a**4)+
     1      a**3+2d0*c*psi*r)/r**2
      return
      end
```